Parabola Volume 19, Issue 3 (1983)

AN APPLESOFT PROGRAM TO EVALUATE PI

Scot McPhie of Kings School Paramatta, has written asking for help in writing a programme for the Apple computer (using Applesoft basic as the language) to calculate the value of π to as many decimal places as desired. He indicates some areas of difficulty with his own attempts, and suggests that to publish a working programme would be of interest. The following is in response to his letter.

Although Gregory's Series $\frac{\pi}{4} = 1 - \frac{1}{3} + \frac{1}{5} - \frac{1}{7} + \ldots + (-1)^n \frac{1}{2n+1} + \ldots.$

can be used to calculate pi to a few decimal places, the convergence of the series is far too slow for practical purposes. The series quoted is the case $x = 1$ of the more general result

$$\tan^{-1}x = \frac{x}{1} - \frac{x^3}{3} + \frac{x^5}{5} - \frac{x^7}{7} + \ldots. \qquad +(-1)^n \frac{x^{2n+1}}{2n+1} + \ldots.. \qquad (1)$$

where $\tan^{-1} x$ is the number of radians in the acute angle whose tangent is x. Using the formula $\tan(\alpha+\beta) = \frac{\tan\alpha + \tan\beta}{1-\tan\alpha\tan\beta}$, you will be able to confirm that

$\tan\left(4\tan^{-1}\frac{1}{5}\right) = \frac{120}{119} = \tan\left(\frac{\pi}{4} + \tan^{-1}\frac{1}{239}\right)$ and hence that

$$\frac{\pi}{4} = 4\tan^{-1}\frac{1}{5} - \tan^{-1}\frac{1}{239} \qquad (2)$$

The formula (2) in conjunction with the series (1) has been one of the favourite methods used to calculate pi to high accuracy, and is the method on which the following Applesoft programme is based.

In this programme the value of π is eventually obtained in the array A(0,J) J = 0, 1,2,...,N%, where each entry in the array is a block of 5 digits. Thus (A(0,0) is (eventually) 31415, (the first 5 digits of π) and (A(0,1) is 92653 (the next five digits) and so on. During the calculation, the arrays A(1,J) A(2,J), and A(3,J) similarly contain decimal expansions of numbers in 5 digit blocks, the decimal point always belonging after the first digit of the first block (i.e. the block corresponding to J = 0).

The segment of programme from line 1000 to 1100 calculates $\tan^{-1}\frac{1}{5}$. The array A(2,J) contains in succession $\left(\frac{1}{5}\right)^1, \left(\frac{1}{5}\right)^3, \left(\frac{1}{5}\right)^5, \ldots, \left(\frac{1}{5}\right)^{2n+1}$ as

the programme runs, and A(1,J) similarly holds $\left(\frac{1}{5}\right)$, $\frac{1}{3}\left(\frac{1}{5}\right)^3$, $\frac{1}{5}\left(\frac{1}{5}\right)^5$, ... , $\left(\frac{1}{2n+1}\right)\left(\frac{1}{5}\right)^{2n+1}$, ... The contents of A(1,J) are alternately added to or substracted from A(0,J) until enough terms have been taken in the series (1) to achieve the required accuracy. At the completion of this process, the number in A(0,J) is multiplied by 4 (line 1120), and the calculated value of $4 \times \tan^{-1}\frac{1}{5}$ is left there undisturbed until the next segment of the programme (lines 1200 - 1320) is completed. This segment similarly calculates $\tan^{-1}\frac{1}{239}$ in A(1,J) having stored powers of $\frac{1}{239}$ in A(3,J) and terms of (1) in A(2,J) during the calculation. Finally (lines 1400-1410) A(1,J) is subtracted from A(0,J) and A(0,J) is multiplied by 4. It now contains our calculated value of pi, which is printed out in lines 1600 - 1690.

With this programme the apple computer calculates π to 100 places in about 3 minutes, and to 1000 places in about 4 hours. This does not compare very well with the speed of the programmes used on large computers to evaluate π to the order of a million places, which I believe take less than one minutes for 1000 places. I am not sure whether my programme is still very inefficient, or whether the inherent limitations of the 6502 Microprocessor-Applesoft language combination are to be held responsible. In any case, 1000 digits in 4 hours is of course a fantastic achievement compared with the best efforts of pre-computer days.

LIST

```
10   INPUT "NUMBER OF DECIMAL PLACES REQUIRED";N%:N1% = N% / 5 + 1
15   INPUT "HARD COPY PRINT OUT REQUIRED?(Y-N)";W$
20   DIM A(3,N1%)
30   FOR I = 0 TO 2: FOR J = 0 TO N1%:A(I,J) = 0: NEXT J,I
40   HOME : GOTO 1000
```

```
150  REM      SUBRT TO ADD A (S%, ) TO A (U%, )
160  C1% = 0
170  FOR K = N1% TO Q% STEP  - 1
180  X = A(U%,K) + A(S%,K) + C1%
190  C1% = 0:  IF X > 99999 THEN C1% = 1:X = X - 100000
200  A(U%,K) = X
210   NEXT K
215   IF C1% = 1 THEN  STOP
220   RETURN
```

```
230   REM    SUBTRACT A (S%, ) FROM   A (U%, )
240  C2% = 0
250   FOR K = N1% TO Q% STEP  - 1
260  X = A(U%,K) + 100000 - A(S%,K) - C2%
270  C2% = 1:  IF X > 99999 THEN X = X - 100000:C2% = 0
280  A(U%,K) = X
290   NEXT K
295   IF C2% > 0 THEN  STOP
300   RETURN
```

```
310   REM     A (U%, ) ==>P%*A (U%, )
320  C3% = 0
330   FOR K = N1% TO 0 STEP  - 1
340  X = P% * A(U%,K) + C3%
350  C3% = X / 100000
360  A(U%,K) = X - 100000 * C3%
370   NEXT K
375   IF C3% > 0 THEN  STOP
380   RETURN
```

```
390   REM    A (T%, )/Z% ==> A (S%, )
400  K% = Q%
410  X = A(T%,K%)
420  D1 =  INT (X / Z%):A(S%,K%) = D1
430  Y% = X - D1 * Z%
440  K% = K% + 1: IF K% > N1% THEN RETURN
$%)  X = 100000 * Y% + A(T%,K%)
460   GOTO 420
```

```
470  M% = M% - 1
480   IF M% * H1 + LOG (M%) < H2 THEN M% = M% + 1: RETURN
490   GOTO 470
```

```
1000   REM   CALCULATE 4*ATN(1/5) AND STORE IN A%(0,)
1010  H1 =  LOG (5):H2 = N1% * 5 *  LOG (10):M% = H2 / H1 + 1: GOSUB 470
1012   PRINT M%          (M% is the denominator of the last term of series (1)
                          required for the desired accuracy)
1020  SN% = 1:L% = 1:Q% = 0:L5 =  LOG (5) /  LOG (10):T% = 2:S1% = 1:U% = 0
1030  A(2,0) = 2000:A(0,0) = 2000
1040  L% = L% + 2:SN% = - SN%: IF L% > M% THEN 1120
1042   PRINT M%,L%       (L% is the denominator of the term of (1) which is being
                          calculated)
1050  S% = T%:Z% = 25: GOSUB 390
1060  S% = S1%:Z% = L%: GOSUB 390
1070   IF SN% > 0 THEN  GOSUB 150
1080   IF SN% < 0 THEN  GOSUB 230
1090  Q% = L% * L5 / 5
1100   GOTO 1040
```

```
1120  P% = 4: GOSUB 310
```

```
1200  H1 =  LOG (239):M% = H2 / H1 + 1: GOSUB 470
1210  SN% = 1:L% = 1:Q% = 0:L239 = H1 /  LOG (10)
1220  T% = 3:S1% = 2:U% = 1
1230   FOR I = 2 TO 3: FOR J = 0 TO N1%:A(I,J) = 0: NEXT J,I
1240  A(3,0) = 10000:Z% = 239:S% = 3: GOSUB 390
1250   FOR J = 0 TO N1%:A(1,J) = A(3,J): NEXT J
1260  L% = L% + 2:SN% =  - SN%: IF L% > M% THEN 1400
1270  S% = T%:Z% = 239: GOSUB 390: GOSUB 390
1280  S% = S1%:Z% = L%: GOSUB 390
1290   IF SN% > 0 THEN  GOSUB 150
1300   IF SN% < 0 THEN  GOSUB 230
1302   PRINT M%,L%
1310  Q% = L% * L239 / 5
1320   GOTO 1260
```

```
1390   REM   FINALLY SUBTRACT A%(1,) FROM A%(0, ), MULTIPLY BY4, AND PRINT OUT THE
                                                                    ANSWER
1400  S% = 1:U% = 0:Q% = 0: GOSUB 230
```

```
1410  GOSUB 310
1590  IF W$ ="Y" THEN  PR# 1   (The printer is assumed to be controlled by a card
                                in slot #1)
1600  FOR J = 0 TO N1%
1610  FOR K = 0 TO 5
1615  IF 6 * J + K > N1% - 2 THEN 1700
1620 W = A(0,6 * J + K)
1630  IF W > 9999 THEN  PRINT W;" ";: GOTO 1690
1640  PRINT "0";: IF 10 * W > 9999 THEN  PRINT W;" ";: GOTO 1690
1650  PRINT "0";: IF 100 * W > 9999 THEN  PRINT W;" ";: GOTO 1690
1660  PRINT "0";: IF 1000 * W > 9999 THEN  PRINT W;" ";: GOTO 1690
1670  PRINT "0";: PRINT W;
1690  NEXT K: PRINT : NEXT J
1700  END
```

OLYMPIAD OMNIBUS (Continued from page 14)

Problem 0.0.2 and solution.

In a triangle ABC, a point P is chosen on AB such that P is between A and B and is nearer to A than to B. Points Q, R and S on AC, CB and BA respectively are such that PQ is parallel to BC, QR is parallel to AB and RS is parallel to CA. Calculate the maximum value that the area of quadrilateral PQRS could have as a fraction of the area of triangle ABC.

Solution:



Using similar triangles defined by the sets of parallel lines, and with the ratio AP/AB = x say then ΔAPQ = area $\Delta BSR = x^2$ area ΔABC while area $\Delta CQR = (1-x)^2$ area ΔABC.

By subtraction area PQES = $(2x - 3x^2)$ area ΔABC

$2x - 3x^2 = 1/3 - 3(x - 1/3)^2 \geq 1/3$

when $x = 1/3$, $2x - 3x^2 = 1/3$ for a maximum value

$\therefore$ The max value of area of PQRS = 1/3 area of ΔABC.

[See page 26 for problem 0.0.3.]